



**Διαχείριση Φάσεων Κατασκευής  
Construction Stage Manager – CSM  
V. 11.25-21**

**Περιγραφή – Παραδείγματα εφαρμογής  
Μάϊος 2005**

**SOFISTiK Hellas A.E.**

3ης Σεπτεμβρίου 56, 104 33 Αθήνα


Τηλ: 2108220607, 2108251632

Fax: 2108251632

[www.sofistik.gr](http://www.sofistik.gr), [info@sofistik.gr](mailto:info@sofistik.gr)

## ΠΕΡΙΕΧΟΜΕΝΑ

Γενικά .....	3
Πίνακας φάσεων κατασκευής CS_Tab.....	4
Πίνακας των groups GR_Tab .....	5
Επεξήγηση ‘_CSM’ αρχείου – Παράδειγμα csm11_prestressed_single_span.dat .....	6
Αρχείο csm11_prestressed_single_span.plb .....	8
Αρχείο csm11_prestressed_single_span_csm.dat .....	8
Γέφυρα με μέθοδο προβολοδόμησης – Παράδειγμα csm5_free_cantilever.dat .....	12
Υπολογισμός παραμορφώσεων – Παράδειγμα cantilever_simple.dat .....	15
Προκ/μένη γέφυρα με σκυροδέτηση επιτόπου (in-situ) – Παράδειγμα csm3_stab_fertigteil_ob.dat .....	18
Προεντεταμένη κλίνη – Παράδειγμα csm3_spannbett-eng.dat .....	20



Εικόνα 1

### Νέο πρόγραμμα διαχείρισης φάσεων κατασκευής Construction Stage Manager - CSM Γενικά

Το νέο πρόγραμμα CSM διαχειρίζεται προβλήματα φάσεων κατασκευής και ερπυσμού με εύκολο τρόπο. Παράγεται αυτόματα ένα αρχείο δεδομένων για τα προγράμματα ASE και AQB (απαιτείται άδεια χρήσης ASE1).

Μέσω ενός αρχείου δεδομένων του προγράμματος CSM, καθορίζονται οι ιδιότητες των φάσεων κατασκευής, δηλαδή τα groups των στοιχείων που συμμετέχουν σε κάθε φάση, οι παράμετροι υπολογισμού των συντελεστών ερπυσμού, συστολής ξήρανσης και χαλάρωσης και οι φορτίσεις.

Πιο αναλυτικά, σε ένα αρχείο δεδομένων, είναι απαραίτητο να περιγράφονται τα υλικά και οι διατομές (πρόγραμμα AQUA), το αναλυτικό προσομοίωμα (πρόγραμμα GENF), εάν το σύστημα είναι προεντεταμένο, τότε η γεωμετρία των καλωδίων προέντασης και οι παράμετροι που καθορίζουν το σύστημα προέντασης (πρόγραμμα GEOS) και τα πρόσθετα φορτία (πρόγραμμα SOFILOAD). Το αρχείο με το οποίο θα καθορίζεται η αλληλουχία των φάσεων και οι προϋπάρχουσες εντατικές καταστάσεις θα προκύψει αυτόματα από τα δεδομένα του CSM.

Σε ένα αρχείο δεδομένων CSM, οι φάσεις κατασκευής ελέγχονται ως ακολούθως:

- Μέσω της εντολής CS καθορίζονται οι φάσεις κατασκευής της εκάστοτε εφαρμογής. Δημιουργείται ένας πίνακας δεδομένων με τις απαιτούμενες φάσεις.
- Η εντολή GRP καθορίζει σε ποιες φάσεις είναι ενεργό το κάθε group στοιχείων. Δημιουργείται ένας πίνακας δεδομένων με την σχετική περιγραφή.
- Η εντολή LC καθορίζει τα πρόσθετα φορτία. Δημιουργείται ένας πίνακας δεδομένων με τις πρόσθετες φορτίσεις και σε ποια φάση αναφέρονται. Οι βασικές φορτίσεις, όπως ίδιο βάρος, προένταση και ερπυσμός υπολογίζονται αυτόματα από το πρόγραμμα.

Οι τένοντες ενεργοποιούνται αυτόματα, ανάλογα με την χρονική στιγμή της τάνυσης του κάθενος και τον αριθμό της φόρτισης στην οποία είναι αποθηκευμένοι. Δηλαδή, η τιμή της παραμέτρου ICS1, έτσι όπως καθορίζεται από το πρόγραμμα GEOS, πρέπει να ταυτίζεται με τον αριθμό της φάσης στην οποία ενεργοποιείται, καθώς και ο αριθμός της φόρτισης στην οποία αποθηκεύονται. Ειδική περίπτωση αποτελεί η προεντεταμένη κλίνη (δες παράδειγμα 'Προεντεταμένη κλίνη – Παράδειγμα csm3\_spannbett-eng.dat').

Το ίδιο βάρος ενεργοποιείται αυτόματα από τα groups που συμμετέχουν στην συγκεκριμένη φόρτιση.

#### Ονοματολογία:

CSM	Construction Stage Manager.
CS	Φάση κατασκευής. Εάν πρόκειται για φάση προέντασης, τότε δίνουμε το ίδιο νούμερο με τη τιμή της παραμέτρου ICS1 των τενόντων που προεντείνονται στην συγκεκριμένη φάση.
LC	Περίπτωση φόρτισης.

CS_Tab	Πίνακας περιγραφής φάσεων κατασκευής σε ένα αρχείο δεδομένων.
GR_Tab	Πίνακας ομάδων στοιχείων (groups) που συμμετέχουν στη κάθε φάση σε ένα αρχείο δεδομένων.
LC_Tab	Πίνακας φορτίσεων σε ένα αρχείο δεδομένων.

### Προϋποθέσεις

Ένας ολοκληρωμένος υπολογισμός είναι εφικτός, μόνον όταν οι φάσεις κατασκευής, οι φάσεις προέντασης και τα groups, χρησιμοποιηθούν έτσι, ώστε να λαμβάνεται υπόψη μία χρονική αλληλουχία μεταξύ τους.

Για παράδειγμα, πρώτα σκυροδετείται το group 50 (το σκυρόδεμα αποκτά τη πλήρη ακαμψία του στη φάση κατασκευής 50) και ακολούθως, στην φάση κατασκευής CS 51 προεντείνονται οι τένοντες του ίδιου group στοιχείων, που έχουν οριστεί με ICS1= 51.

Για την απλοποίηση και τη τυποποίηση των δεδομένων και αποτελεσμάτων, χρησιμοποιούνται αριθμοί φορτίσεων τέτοιοι, ώστε να ταιριάζουν με τον ίδιο αριθμό της φάσης κατασκευής.

Π.χ. στο αρχείο που παράγεται αυτόματα, τα δεδομένα για το πρόγραμμα ASE για την φάση 10 είναι τα ακόλουθα:

```

PROG ASE
HEAD Construction stage CS 10 Τίτλο βάρος προκατασκευής
SYST PLC 0 $(ASE_SYST)
GRP 10 CS 10
GRP 91 CS 10
LC 4010 TYPE G_1 DLZ 1.0 TITL 'Τίτλο βάρος προκατασκευής'
END

```

Η φόρτιση 4010 αντιστοιχεί στην φάση 10 και αποτελείται από το ίδιο βάρος των στοιχείων που ανήκουν στα groups 10 και 91.

Κάτω από αυτές τις προϋποθέσεις, μέσω του Construction Stage Manager παράγεται γρήγορα ένα αρχείο δεδομένων για το σύστημα, όπου περιγράφεται η ακολουθία των φάσεων κατασκευής.

### Πίνακας φάσεων κατασκευής CS\_Tab

Αυτός ο πρώτος πίνακας, σε ένα αρχείο δεδομένων, περιγράφει την αλληλουχία των φάσεων κατασκευής στο προς επίλυση σύστημα. Η εντολή που χρησιμοποιείται είναι η ακόλουθη:

**CS: NO – TYPE – T – RH – TEMP – NCRE – TITL**

<b>TYPE:</b>	<b>Είδος φάσης. Έχει σχέση με το είδος της φόρτισης που θα παραχθεί για την αντίστοιχη φάση.</b>	
<b>G_1 ή G</b>	Σκλήρυνση ενός νεοσυσταθέντος τμήματος σκυροδέματος ή εισαγωγή ενός στοιχείου (ελατηρίου, καλωδίου...). Ενεργοποιείται αυτόματα το ίδιο βάρος.	
<b>C</b>	Φάση ερπυσμού με διάρκεια επίδρασης στο σκυρόδεμα εξαρτώμενη από τη θερμοκρασία.	
<b>P</b>	Προένταση. Η φάση αυτή αντιστοιχεί στην χρονική στιγμή ICS1 για κάθε τένοντα, όπως καθορίζεται στο πρόγραμμα GEOS, εκτός από τη περίπτωση της προεντεταμένης κλίνης (δες παράδειγμα 'Προεντεταμένη κλίνη – Παράδειγμα csm3_spannbett-eng.dat').	
<b>SL</b>	Βραχυπρόθεσμο κινητό φορτίο σε τμήμα του συστήματος χωρίς επίδραση στον ερπυσμό (το κινητό φορτίο απομακρύνεται αμέσως, δεν αποτελεί προϋπάρχουσα εντατική κατάσταση για την επόμενη φόρτιση).	
<b>G_2</b>	Τοποθέτηση ενός μόνιμου φορτίου με επίδραση στον ερπυσμό. Αποτελεί προϋπάρχουσα εντατική κατάσταση για την επόμενη φόρτιση. Ένα τέτοιο φορτίο πρέπει να αναφέρεται σε μία δική του φάση (CS), αλλιώς θα συμπεριληφθεί με άλλα φορτία και δε θα μπορεί να διαχωριστεί κατά την δημιουργία των περιβαλλουσών στα MAXIMA-AQB.	
<b>B</b>	Φάση κατασκευής, π.χ. τοποθέτηση ενός φορτίου συναρμολόγησης με επίδραση στον ερπυσμό, ή φόρτιση συναρμολόγησης π.χ. απομάκρυνση βοηθητικών στηρίξεων. Είναι φόρτιση με μεγαλύτερη διάρκεια π.χ. φορτίο από όχημα προώθησης. Απαιτείται ίδια φάση κατασκευής.	

<b>T*</b>	Ενεργός ηλικία σκυροδέματος στην συγκεκριμένη φάση, για τον υπολογισμό του συντελεστή ερπυσμού και συστολής ξήρανσης.
<b>RH*</b>	Σχετική υγρασία περιβάλλοντος για τον υπολογισμό του συντελεστή ερπυσμού.
<b>TEMP</b>	Μέση θερμοκρασία περιβάλλοντος για τον υπολογισμό του συντελεστή ερπυσμού.
<b>NCRE</b>	Βήματα υπολογισμού συντελεστή ερπυσμού, ώστε το delta-phi ενός βήματος να μην μεγαλώσει πολύ (μία καλή εκτίμηση είναι μέγιστη τιμή phi=0.4).

(\*) Για τις φάσεις ερπυσμού δίνεται η χρονική διάρκεια της φάσης, η θερμοκρασία περιβάλλοντος και σχετική υγρασία. Τα δεδομένα αυτά αφορούν όλα τα στοιχεία που συμμετέχουν στην φάση (αν για παράδειγμα πρέπει να δοθούν διαφορετικές τιμές υγρασίας για κάθε group, πρέπει αυτό να τεθεί εκ των υστέρων στη βάση δεδομένων).

Πίνακας 1

**ΠΡΟΣΟΧΗ! Φάση κατασκευής 0 δεν επιτρέπεται να χρησιμοποιείται.**

### Πίνακας των groups GR\_Tab

Σε ένα αρχείο δεδομένων, στον πίνακα των groups περιγράφονται τα Groups των στοιχείων που συμμετέχουν σε κάθε φάση και με ποιές ιδιότητες είναι ενεργοποιημένο.

Ισχύουν οι ακόλουθες εντολές:

**GRP: NO – ICS1 – ATIL – HFIX – BEDD – SITU – T0 – FAC1 – PHIF – QUEA**

<b>ICS1</b>	Αριθμός φάσης κατασκευής στην οποία η ακαμψία του group ενεργοποιείται για πρώτη φορά. Μέσα στη διατομή μπορούν να ενεργοποιηθούν τμήματα σε επόμενη φάση. Αυτό καθορίζεται μέσω του προγράμματος AQUA. Default: ICS1 = αριθμός group
<b>ATIL</b>	Φάση κατά την οποία απομακρύνεται κάποιο group στοιχείων που μέχρι τώρα ήταν ενεργό (π.χ. ελατήρια βοηθητικών στηρίξεων). Default: ATIL = 999
<b>HFIX</b>	Αριθμός φάσης κατασκευής κατά την οποία δεσμεύονται (καταργούνται) οι αρθρώσεις των ραβδωτών στοιχείων. Default: Οι αρθρώσεις είναι πάντα ενεργές.
<b>BEDD</b>	Φάση κατασκευής κατά την οποία ενεργοποιείται για το συγκεκριμένο group στοιχείων η επιφανειακά κατανεμημένη ελαστική έδραση (bedding σε QUAD-στοιχεία). Default: Η ελαστική έδραση είναι πάντα ενεργή
<b>SITU*</b>	Φάση κατασκευής κατά την οποία ενεργοποιείται για το συγκεκριμένο group στοιχείων η επιτόπου σκυροδέτηση (in-situ). Default: CS
<b>T0 (μηδέν)</b>	Ενεργός ηλικία σκυροδέματος κατά την επιβολή της πρώτης τάσης – σε ημέρες. Default: 7 ημέρες
<b>FAC1</b>	Συντελεστής ακαμψίας για τη πρώτη ενεργοποίηση group.
<b>PHI1</b>	Συντελεστής ερπυσμού για ελατήρια και θεμελιώσεις.
<b>QUEA</b>	Αξονική ακαμψία QUAD-στοιχείων (τροποποιείται το E*A των στοιχείων).

(\*) Σε συστήματα όπου σε κάποια φάση προβλέπεται επιτόπου σκυροδέτηση, ορίζεται μέσω της παραμέτρου SITU η φάση στην οποία θα υπολογιστεί το ίδιο βάρος νωπού σκυροδέματος (Ortbeton) τμημάτων μεταγενέστερων φάσεων κατασκευής, χωρίς να είναι ακόμα ενεργή η ακαμψία τους. Βλέπε παράδειγμα 'CSM - Δοκός δύο ανοιγμάτων με προκ/νες δοκούς και επί τόπου σκυροδέτηση' (csm3\_stab\_fertigteil\_ob.dat).

Πίνακας 2

## Πίνακας των φορτίσεων LC\_Tab

Με τον πίνακα φορτίσεων καθορίζονται οι φάσεις κατασκευής στις οποίες ενεργοποιούνται οι πρόσθετες φορτίσεις (έχουν περιγραφεί στο πρόγραμμα SOFILOAD). Ισχύουν οι ακόλουθες εντολές:


**LC: NO – TYPE – ICS1 – ATIL**

<b>TYPE:</b>	όπως TYPE της εντολής CS
<b>ICS1:</b>	Φάση κατά την οποία αρχίζει να δρα η φόρτιση. Default: NO = αριθμός φόρτισης
<b>ATIL:</b>	Φάση κατά την οποία παύει να δρα η συγκεκριμένη φόρτιση. Default: δρα επ' άπειρον

Πίνακας 3

## Επεξήγηση '\_CSM' αρχείου – Παράδειγμα csm11\_prestressed\_single\_span.dat

Πρόκειται για ένα παράδειγμα επίτηδες απλοποιημένο, για να φανεί η χρήση του προγράμματος. Είναι μία προεντεταμένη δοκός ενός ανοίγματος, ορθογωνικής διατομής με ευθύγραμμους τένοντες προέντασης.



Εικόνα 2

Τα δεδομένα του προγράμματος GEOS είναι τα ακόλουθα:

```

PROG GEOS
HEAD PRESTRESS
CBEA NOH 1 FROM 1 TO 20 TYPE BEAM
TGEO NOG 1 NOH 1 TITL 'STRAIGHT TENDON'
PTSC NO X U V DVS TYPE KIND
      1 0 0 0.65 0 0 PRES
      11 0 0 0.65 0 0 -
      20 0.5 0 0.65 0 0 FIX

PSYS MNO 1002 ZV 250 AZ 179.2 EZ 170000 MUE 0
PRES NOT 11 NOG 1 ICS1 11 ICS2 12
PFAC KIND LE ANWS 9 FAC 1 FACA 1 FACS 1
TEND NTEN 5

PSYS MNO 1002 ZV 250 AZ 179.2 EZ 170000 MUE 0
PRES NOT 21 NOG 1 ICS1 21 ICS2 22
PFAC KIND LE ANWS 9 FAC 1 FACA 1 FACS 1
TEND NTEN 5

PSYS MNO 1002 ZV 250 AZ 179.2 EZ 170000 MUE 0
PRES NOT 31 NOG 1 ICS1 31 ICS2 32
PFAC KIND LE ANWS 9 FAC 1 FACA 1 FACS 1
TEND NTEN 5

LOAD NOT 11 LC 11
LOAD NOT 21 LC 21
LOAD NOT 31 LC 31
END

```

Οι τένοντες με αριθμό 11 προεντίνονται την χρονική στιγμή 11, αποκτούν συνάφεια την χρονική στιγμή 12 και αποθηκεύονται στην φόρτιση με αριθμό 11 (PRES NOT 11, ....ICS1 11 ICS2 12 και LOAD NOT 11 LC 11). Ο αριθμός της φόρτισης πρέπει να ταυτίζεται με τον αριθμό της φάσης στην οποία θα ενεργοποιηθεί η συγκεκριμένη προένταση.

Αντίστοιχα, οι τένοντες με αριθμό 21 προεντίνονται την χρονική στιγμή 21, αποκτούν συνάφεια την χρονική στιγμή 22 και αποθηκεύονται στην φόρτιση με αριθμό 21, οι τένοντες με αριθμό 31 προεντίνονται την χρονική στιγμή 31, αποκτούν συνάφεια την χρονική στιγμή 32 και αποθηκεύονται στην φόρτιση με αριθμό 31.

Οι πιο πάνω τιμές πρέπει να ταυτίζονται με τον αριθμό της φάσης κατασκευής που θα πραγματοποιηθεί η κάθε προένταση, δηλ. στην CS 11 θα ενεργοποιηθεί η προένταση που πραγματοποιείται την χρονική στιγμή 11, στην CS 21 αυτή της χρονικής στιγμής 21 και στην CS 31 αυτή της χρονικής στιγμής 31.

Τα δεδομένα για το πρόγραμμα CSM είναι τα ακόλουθα:

Πίνακας φάσεων κατασκευής

CS	NO	TYPE	T(d)	NCRE	
	10	G_1			Υπολογισμός του ίδιου βάρους της δοκού.
	11	P			Υπολογισμός προέντασης τενόντων με ICS1=11.
	15	C	10		Υπολογισμός πρώτου βήματος ερπυσμού με χρονική διάρκεια T= 10 ημερών.
	21	P			Υπολογισμός προέντασης τενόντων με ICS1=21
	25	C	10		Υπολογισμός δεύτερου βήματος ερπυσμού με χρονική διάρκεια T= 10 ημερών.
	31	P			Υπολογισμός προέντασης τενόντων με ICS1=31
	35	C	10		Υπολογισμός τρίτου βήματος ερπυσμού με χρονική διάρκεια T= 10 ημερών.
	50	G_2			Τοποθέτηση ασφαλτικών και γενικά πρόσθετων μόνιμων φορτίων.
	55	C	50		Υπολογισμός ερπυσμού με χρονική διάρκεια T= 50 ημερών.
	65	C	30000	5	Υπολογισμός ερπυσμού σε χρόνο άπειρο. Εδώ ζητείται από το πρόγραμμα να κάνει τον υπολογισμό σε 5 βήματα, για μεγαλύτερη ακρίβεια (NCRE = 5).

\*\*\* Για τις φάσεις ερπυσμού εφόσον δεν δίνεται η θερμοκρασία περιβάλλοντος και σχετική υγρασία, θα ληφθούν υπόψη οι 'default' τιμές, που είναι: RH=70% και TEMP=20°.

Πίνακας 4

Πίνακας Groups στοιχείων

GRP	NO	ICS1	T0	
	0	10	28	Όλος ο φορέας ενεργοποιείται ταυτόχρονα στην φάση 10 (από την αρχή). Για τον υπολογισμό του συντελεστή ερπυσμού, χρειάζεται ο καθορισμός της ενεργούς ηλικίας σκυροδέματος κατά την επιβολή της τάσης (ξεκαλούπωμα) για αυτό καθορίζεται ο χρόνος T0.

Πίνακας 5

Πίνακας φορτίσεων

LC	NO	TYPE	ICS1	
	2	G_2	50	Πρόσθετα μόνιμα φορτία. Ενεργοποιούνται στην φάση κατασκευής 50 (ICS1).

Πίνακας 6

Με τα μέχρι εδώ δεδομένα, το πρόγραμμα θα υπολογίσει αυτόματα τους συντελεστές ερπυσμού και συστολής ξήρανσης, για κάθε φάση. Το άθροισμα των επιμέρους συντελεστών, αποτελεί τον συνολικό συντελεστή ερπυσμού και συστολής ξήρανσης, που θα χρησιμοποιηθεί για τον υπολογισμό των απωλειών τάσεων.

Στη περίπτωση που ο χρήστης επιθυμεί να προσδιορίσει τον συνολικό συντελεστή ερπυσμού και συστολής ξήρανσης, χρησιμοποιεί την εντολή CREP και το πρόγραμμα κατανέμει τη τιμή αυτή στις διάφορες φάσεις, ανάλογα με το χρονικό διάστημα που μεσολαβεί από την μία στην άλλη:

CREP MNO 1 GRP – PHI 2.20 EPS –30E-5





από το AQB και υπολογίζει τα υπερστατικά μέρη στα ραβδωτά στοιχεία, καθώς και τις συνολικές απώλειες ερπυσμού στα επιφανειακά πεπερασμένα στοιχεία – εάν υπάρχουν.

Κατά τον υπολογισμό με το πρόγραμμα AQB, παράγονται οι φορτίσεις με αρίθμηση ίση με 6000+(τον αριθμό της φάσης στην οποία αναφέρονται).

Τέλος, στο αρχείο αυτό περιλαμβάνονται συνδυασμοί των πιο πάνω φορτίσεων, για τον υπολογισμό των τάσεων κάθε φάσης. Αυτοί οι συνδυασμοί έχουν αρίθμηση ίση με 7000+(τον αριθμό της φάσης στην οποία αναφέρονται).

#### OVERVIEW LOADCASES

Total CS displacements and forces starting at loadcase number. 4000

Difference displacements and forces starting at loadcase number. 5000

AQB inner stresses from creep and shrinkage from loadcasenumber. 6000

Stress results of the AQB-LCST-evaluation from loadcasenumber. 7000

### Επεξήγηση εντολών:

Γενικό μέρος αρχείου με 'blocks' δεδομένων, για να αποφεύγεται η επανάληψη ίδιων εντολών.

```
#define BEAM_NO=      11
#define BEAM__X=      0.0000
```

Επιλογή στοιχείων και διατομών, όπου θα γίνει υπολογισμός τάσεων. Για να ληφθούν υπόψη όλα τα στοιχεία, δίνουμε στις παραμέτρους τιμή '—' (παύλα). Η επανάληψη των παραμέτρων αυτών γίνεται με:  
\$(BEAM\_NO) και \$(BEAM\_X)

```
#define ASE_CTRL
ECHO disp, reac, forc, nost, bedd no
CTRL DIFF 1000 $ difference loadcases increment
NSTR S0 KSV SL
CTRL CANT 2      $ 2 = tangential free cantilever
                  $ 1 = with original inclination
#enddef
```

Ομάδα γενικών εντολών του ASE. Όταν CTRL DIFF = 1000, τότε οι φορτίσεις που αναφέρονται στην διαφορά της φάσης από την προϋπάρχουσα έχουν αρίθμηση 5000. Η επανάληψη των εντολών αυτών γίνεται με:

#include ASE\_CTRL

```
#define ASE_SYST=prob line $ nonl nmat yes iter 60
```

Σε περίπτωση που χρειάζεται να γίνει μη γραμμική ανάλυση, πρέπει να τροποποιηθούν οι τιμές των παραμέτρων της εντολής SYST. Η επανάληψη της παραμέτρου αυτής γίνεται με:  
\$(ASE\_SYST)

```
#define AQB_CTRL
ECHO stre no      ; ECHO eige no
CTRL EIGE 1+2
#enddef
```

Ομάδα γενικών εντολών του AQB. Με CTRL EIGE 1+2 καθορίζεται η μέθοδος υπολογισμού των απωλειών τάσεων λόγω ερπυσμού – δεξ εγχειρίδιο AQB. Η επανάληψη των εντολών αυτών γίνεται με:  
#include AQB\_CTRL

```
#define AQB_BEAM
BEAM CS0  11 CS1  12 CS2  21 CS3  22 CS4  31 CS5  32
#enddef
```

Εντολή BEAM προγράμματος AQB. Αναφέρονται οι φάσεις διατομής, όπως έχουν αποθηκευθεί στην βάση δεδομένων από το CSM. Η επανάληψη της εντολής αυτής γίνεται με:  
#include AQB\_BEAM

Στο πρόγραμμα AQB, από την έκδοση V 12.50-21, μέσω της εντολής BEAM, επιτρέπονται μέχρι 99 φάσεις κατασκευής στην ίδια διατομή. Στη περίπτωση που οι φάσεις είναι περισσότερες των 10, τότε η σύνταξη των δεδομένων γίνεται ως εξής:

Για κάθε δεκάδα φάσεων που προσθέτουμε, σαν τιμή της παραμέτρου FROM γράφουμε 'CS+X' και ορίζουμε τις 10 επόμενες φάσεις. Παρ' όλο που η ονομασία των παραμέτρων ξεκινά πάλι από CS0, CS1, ... τους δίνεται αρίθμηση +10 από τη προηγούμενη σειρά.

```

BEAM   CS0   11   12   21   22   31   32   41   42   51   52   $ φάσεις από 0 - 9
'CS+X' CS0   61   62   71   72   81   82   91   92  101  102   $ φάσεις από 10 - 19
'CS+X' CS0  111  112  121  122  131  132  141  142  151  152   $ φάσεις από 20 - 29
'CS+X' CS0  161  162                                     $ φάσεις από 30 - 31

```

Δεδομένα φάσης κατασκευής 10:

```

PROG ASE
HEAD Construction stage CS 10 g1
#include ASE_CTRL
SYST PLC 0 $(ASE_SYST)
GRP 0 CS 10
LC 4010 TYPE G_1 DLZ 1.0 TITL 'g1'
END

```

Η φάση κατασκευής 10 περιλαμβάνει το ίδιο βάρος του group 0. Λαμβάνει υπόψη σαν προϋπάρχουσα εντατική κατάσταση τη PLC 0 (δηλαδή καμιά, εφόσον είναι η πρώτη φόρτιση) και οι φορτίσεις που προκύπτουν είναι οι 4010 και 5010 (CTRL DIFF 1000). Σ' αυτή τη πρώτη φάση, οι δύο αυτές φορτίσεις ταυτίζονται, διότι δεν υπάρχει προϋπάρχουσα εντατική κατάσταση PLC.

Φάση κατασκευής 11:

```

PROG ASE
HEAD Construction stage CS 11 prestress
#include ASE_CTRL
SYST PLC 4010 $(ASE_SYST)
GRP 0 CS 11
LC 4011 TYPE V DLZ 1.0 TITL 'prestress'
LCC 11 PLC NEW $ acting the first time
END

```

Την χρονική στιγμή 11 ενεργοποιείται η προένταση των τενόντων 11 (δες και δεδομένα GEOS). Στην φάση κατασκευής 11, λαμβάνεται σαν PLC η φόρτιση 4010 (αμέσως προηγούμενη). Τα φορτία που περιέχονται στη νέα φόρτιση (LC 4011) είναι η προένταση των τενόντων με αριθμό 11 και όλα όσα περιείχε η PLC\*, δηλ. το ίδιο βάρος από τη 4010.

Ταυτόχρονα, παράγεται η 5011 που περιέχει την διαφορά έντασης και παρ/σης από τη 4011 στη 4010 (εδώ από προένταση).

(\*) Σε κάθε νέα φόρτιση πρέπει να επαναλαμβάνονται όλα τα φορτία της PLC, εκτός από αυτεντατικές καταστάσεις, όπως ερπυσμός και θερμοκρασιακή μεταβολή, που μεταφέρονται αυτόματα μέσω της εντολής SYST PLC. Για αυτό είναι χρήσιμη η παράμετρος PLC στην εντολή LCC. Τη πρώτη φορά που ενεργοποιείται μία φόρτιση παίρνει τη τιμή NEW (LCC 11 PLC NEW), ενώ σε όλες τις επόμενες φάσεις όπου επαναλαμβάνεται η φόρτιση, η παράμετρος PLC παίρνει τη τιμή YES (LCC 11 PLC YES).

Φάση κατασκευής 15:

```

PROG AQB
HEAD creep step CS 15 creep
#include AQB_CTRL
#include AQB_BEAM
LC 5010 TYPE 'G_1' CST CS0 REF 'PART' SUP PERM
LC 5011 TYPE 'V' CST CS0 REF 'PART' SUP PERM $ V=prestress
COMB SUM LC1 G 1 V 1 LCST 6015 CST CS1 TITL '15 creep'
EIGE GRP 0 MNO 1 PHI 0.586 EPS -0.0000141
EIGE MNO 2 T 10 $ activates tendon steel relaxation
END

```

Η φάση 15 είναι ο υπολογισμός του πρώτου βήματος ερπυσμού μετά τη πρώτη προένταση της φάσης 11. Τα ισοστατικά μέρη του ερπυσμού παράγονται από το AQB (CTRL EIGE 1). Οι φορτίσεις που λαμβάνονται υπόψη στον συνδυασμό που προκαλεί τον ερπυσμό είναι οι 5010 και 5011\*. Παράγεται η φόρτιση 6015. Οι συντελεστές ερπυσμού και συστολής ξήρανσης που χρησιμοποιούνται είναι αυτοί που υπολογίστηκαν από το πρόγραμμα CSM (δες αρχείο \$(name).plb).

(\*) Σε κάθε AQB χρησιμοποιούνται οι φορτίσεις με αρίθμηση 5000 και 6000 και όχι αυτές με 4000. Ο λόγος είναι ότι εάν στους συνδυασμούς που δημιουργούνται στο AQB χρησιμοποιούσαμε τις φορτίσεις 4000, θα επαναλαμβάνονταν πολλές φορές τα ίδια φορτία, εφόσον περιέχονται στην συνολική ένταση κάθε φάσης.

```

PROG ASE
HEAD Construction stage CS 15 creep
#include ASE_CTRL
SYST PLC 4011 $(ASE_SYST)
CREP 1 T 10 BEAM AQB
GRP 0 CS 15
LC 4015 TYPE C DLZ 1.0 TITL 'creep'
LCC 11 PLC YES $ already applied in PLC
LCC 6015 PLC NEW $ AQB beam creep curvatures
END

```

Τα ισοστατικά μέρη του ερπυσμού που προέκυψαν από το πιο πάνω AQB (φόρτιση 6015), παραλαμβάνονται από το ASE (CREP ... BEAM AQB και LCC 6015 PLC NEW) για τον υπολογισμό των υπερστατικών μερών, οπότε προκύπτει η φόρτιση 4015. Επειδή έχει χρησιμοποιηθεί η 4011 σαν PLC, επαναλαμβάνονται όλα τα φορτία της, δηλ. το ίδιο βάρος και η προένταση. Η φόρτιση 5015 (CTRL DIFF 1000) θα περιέχει την διαφορά της έντασης από τη 4015 στη 4011.

Σε περίπτωση που ο φορέας αποτελείται και από επιφανειακά πεπερασμένα στοιχεία, το πρόγραμμα ASE υπολογίζει εκ νέου τον συντελεστή ερπυσμού. Γι' αυτό, στην εντολή CREP ορίζεται εκ νέου η ηλικία του σκυροδέματος T=10. Η φόρτιση 4015 περιέχει τις απώλειες τάσεων σε αυτά τα στοιχεία.

Οι συνολικές απώλειες τάσεων στον φορέα, μέχρι την φάση 15, είναι το άθροισμα των φορτίσεων 5015 και 6015. Οι συνολικές απώλειες τάσεων στον φορέα, μέχρι την φάση 69 (σε χρόνο άπειρο), είναι το άθροισμα των φορτίσεων 5015+ 6015+5025+6025+5035+6035+5055+6055+5065+6065+...+5069+6069.

Τα δεδομένα για τις υπόλοιπες φάσεις δίνονται με την ίδια μεθοδολογία.

Στο τέλος του αρχείου υπάρχουν δεδομένα για γραφική απεικόνιση των αποτελεσμάτων κάθε φάσης, καθώς και για την δημιουργία συνδυασμών φορτίσεων που θα χρησιμοποιηθούν στους ελέγχους τάσεων και την διαστασιολόγηση:

```

PROG AQB
HEAD CSM Print of stresses in a cross section
$ Please look for "Analysis of Section" in the print file
ECHO SECT,LC,COMB,FORC,REIN,STRE NO
ECHO TABS 2 $ 1+2 = print all tendons separately
BEAM $(BEAM_NO) X $(BEAM_X) CS0 11 CS1 12 CS2 21 CS3 22 CS4 31 CS5 32
$
#define loadcases
LC 5010 TYPE 'G_1' \ CST CS0 REF PART SUP PERM
LC 5011 TYPE 'V' \ CST CS0 REF PART SUP PERM
LC 5015 TYPE 'C' \ CST CS1 REF PART SUP PERM
LC 6015 TYPE 'C' \ CST CS1 REF PART SUP PERM
LC 5021 TYPE 'V' \ CST CS2 REF PART SUP PERM
LC 5025 TYPE 'C' \ CST CS3 REF PART SUP PERM
LC 6025 TYPE 'C' \ CST CS3 REF PART SUP PERM
LC 5031 TYPE 'V' \ CST CS4 REF PART SUP PERM
LC 5035 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6035 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5050 TYPE 'G_2' \ CST CS5 REF PART SUP PERM
LC 5055 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6055 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5065 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6065 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5066 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6066 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5067 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6067 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5068 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6068 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 5069 TYPE 'C' \ CST CS5 REF PART SUP PERM
LC 6069 TYPE 'C' \ CST CS5 REF PART SUP PERM
$ Comparison loadcases - cast in one:
#endif
#include loadcases

```

Καθορισμός των στοιχείων που θα γίνει υπολογισμός τάσεων. Λαμβάνονται υπόψη οι τιμές που δόθηκαν κατά τον ορισμό των παραμέτρων **BEAM\_NO** και **BEAM\_X**. Για επιλογή όλων των στοιχείων, δίνουμε τιμή '-' (παύλα).


Συνολικός πίνακας με τις φορτίσεις που έχουν ήδη υπολογιστεί στις διάφορες φάσεις. Μπορεί ο χρήστης να τον χρησιμοποιήσει για να δημιουργήσει επιπλέον συνδυασμούς φορτίσεων με σκοπό τον έλεγχο τάσεων και την διαστασιολόγηση του φορέα. Η επανάληψη του πίνακα γίνεται με την εντολή: **#include loadcases**

Συνδυασμοί φορτίσεων που έχουν ήδη αποθηκευθεί στην βάση δεδομένων. Ο χρήστης μπορεί να συμπληρώσει σε αυτό το σημείο το αρχείο δεδομένων με τους επιπλέον συνδυασμούς που χρειάζονται για τους ελέγχους τάσεων και διαστασιολόγηση. Υπάρχει επίσης η δυνατότητα ένα αντίστοιχο αρχείο διαστασιολόγησης να προκύψει αυτόματα από το CSM.

```
$ stresses in sum of loadcases:
COMB SUM LC1 5010 F1 .000001 TITL ***** $ creates separation line in listing
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7010 TITL='g1'
      SUM 5010
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7011 TITL='prestress'
      SUM 5010 5011
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7015 TITL='creep'
      SUM 5010 5011 5015 6015
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7021 TITL='prestress'
      SUM 5010 5011 5015 6015 5021
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7025 TITL='creep'
      SUM 5010 5011 5015 6015 5021 5025
      AND 6025
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7031 TITL='prestress'
      SUM 5010 5011 5015 6015 5021 5025
      AND 6025 5031
COMB EXTR LC1 LC2 LC3 LC4 LC5 LC6 LCST=7035 TITL='creep'
      SUM 5010 5011 5015 6015 5021 5025
      AND 6025 5031 5035 6035
....
```

Με το αρχείο αυτό γίνεται πλήρης διαχείριση των φάσεων κατασκευής. Τα δεδομένα που αφορούν στατικούς ή δυναμικούς υπολογισμούς και αναφέρονται στο πλήρες σύστημα (μετά το πέρας των φάσεων κατασκευής), θα προστεθούν από τον χρήστη (σε πλήρες σύστημα, χωρίς χρήση προϋπαρχουσών εντατικών καταστάσεων).

### Γέφυρα με μέθοδο προβολοδόμησης – Παράδειγμα csm5\_free\_cantilever.dat



Εικόνα 3

Κατά την μόρφωση του φορέα, φροντίζουμε να διαχωρίσουμε τα στοιχεία σε groups, σύμφωνα με τη πορεία κατασκευής τους. Η διατομή μεταβάλλεται κατά μήκος των στοιχείων, έτσι ώστε η ενισχυμένη να βρίσκεται στη περιοχή των βάθρων.

Τα δεδομένα του προγράμματος GEOS είναι τα ακόλουθα:

```
PROG GEOS
ECHO FULL NO
CBEA NOH FROM To INC TYPE
      1 117 111 -1 NODE $ DEFINITION
      1 111 10 -101 NODE
      1 10 17 1 NODE
TGEO NOG 1 NOH 1 $ Pure Geometry
let#1 0.15
PTSC NO X U V
      12006 0.000 0.0 #1
      0.000 0.0 #1
      11006 6.000 0.0 #1
$.....
LET#99 12+12 ; PRES NOT 101 NOG 1 ICS1 101 102
PSYS 1003 146*#99 150*#99 195000. 4.8 0.3 0.21 0.21 3. 97. 150*#99 TITL '2 x 12+12'
```

```

PFAC RI  FACS 1.00 ; TEND 111 11 NODE NTEN  2
let#1 111,12  $ First tendon - right
let#3 112,12  $ First tendon - left
let#5 1       $ CS-Number
$.....
#define tendon
let#IBA1 101+#5*10
LET#99 12+12 ; PRES NOT #IBA1 NOG 1  ICS1 #IBA1 #IBA1+1
PSYS 1003 146*#99 150*#99 195000. 4.8 0.3 0.21 0.21  3. 97. 150*#99 TITL  '2 x 12+12'
PFAC RI  FACS 1.00 ; TEND #1 #2 NODE NTEN  2
let#5 #5+1
let#IBA1 101+#5*10
LET#99 12+12 ; PRES NOT #IBA1 NOG 1  ICS1 #IBA1 #IBA1+1
PSYS 1003 146*#99 150*#99 195000. 4.8 0.3 0.21 0.21  3. 97. 150*#99 TITL  '2 x 12+12'
PFAC RI  FACS 1.00 ; TEND #3 #4 NODE NTEN  2
let#5 #5+1
let#1 #1+1
let#2 #2+1
let#3 #3+1
let#4 #4+1
#enddef
#include tendon $ cs 1+2
#include tendon $ cs 3+4
#include tendon $ cs 5+6
#include tendon $ cs 7+8
#include tendon $ cs 9+10
#include tendon $ cs 11+12
$.....Loadcases.....
ECHO ZP0 YES
LOAD NOT LC
      (101 #IBA1 10) (101 10)  $ LC 101 to Loadcase 221
END

```

Για να αποφευχθεί ο μεγάλος όγκος δεδομένων, χρησιμοποιείται παραμετρική εισαγωγή. Περιγράφονται οι τένοντες που προεντίνονται διαδοχικά στις φάσεις 101, 111, 121 έως 221. Αντίστοιχα αποθηκεύονται σε φορτίσεις με αρίθμηση 101, 111, 121 έως 221.

Οι πρόσθετες φορτίσεις κατά το στάδιο κατασκευής προέρχονται από το φορτίο του άδειου φορείου πριν την σκυροδέτηση κάθε νέας φάσης και από το φορτίο του νωπού σκυροδέματος κάθε νέας φάσης, πριν να αποκτήσει τη πλήρη ακαμψία ο νέος σπόνδυλος. Περιγράφονται μέσω του προγράμματος SOFiLOAD:

```

PROG SOFILOAD
let#1 0
      LC 700+#1      TITL  'Empty framework  CS #1'
      BEAM 110,10 TYPE PZZ 100
      LC 800+#1      TITL  'Concrete CS #1'
      BEAM 110,10 TYPE PZZ 10
loop#loop 12
  let#1 1+#loop
    LC 700+#1      TITL  'Empty framework CS #1'
    BEAM #1*1000+1+#1/2-.75 TYPE PZZ 100
    LC 800+#1      TITL  'Concrete CS #1'
    BEAM #1*1000+1+#1/2-.75 TYPE PZZ 10
endloop
END

```

**Πίνακας φάσεων κατασκευής**

CS	NO	TYPE	T(d)	NCRE	
	100	G_1	-		Σκυροδέτηση βάθρων (ενεργοποίηση ι.β.)
	101	P			Προένταση περιοχής βάθρων
	105	C	50	1	Φάση ερπυσμού με χρονική διάρκεια T= 50 ημερών
	109	B			Φορτίο ρύγχους 1
	110	G_1	-		Σκυροδέτηση CS 1 (ενεργοποίηση ι.β.)
	111	P	-		Προένταση CS 1
	115	C	14	1	Φάση ερπυσμού με χρονική διάρκεια T= 14 ημερών
	119	B	-		Φορτίο ρύγχους 2. Ταυτόχρονα απομακρύνεται το προηγούμενο φορτίο ρύγχους προώθησης, μιας και αυτή η πρώτη περίπτωση φόρτισης ρύγχους της φάσης 109 επιδρά μόνο μέχρι τη CS 115. Δες πίνακα φορτίσεων LC_Tab.
	120	G_1	-		Σκυροδέτηση CS 2 (ενεργοποίηση ι.β.)
	121	P	-		Προένταση CS 2
	125	C	14	1	Φάση ερπυσμού με χρονική διάρκεια T= 14 ημερών
	129	B	-		Φορτίο ρύγχους 3. Ταυτόχρονα απομακρύνεται το προηγούμενο φορτίο ρύγχους προώθησης, μιας και αυτή η πρώτη περίπτωση φόρτισης ρύγχους της φάσης 119 επιδρά μόνο μέχρι τη CS 125. Δες πίνακα φορτίσεων LC_Tab.
	130	G_1	-		Σκυροδέτηση CS 3 (ενεργοποίηση ι.β.)
	131	P	-		Προένταση CS 3
	135	C	14	1	Φάση ερπυσμού με χρονική διάρκεια T= 14 ημερών
	139	B	-		Φορτίο ρύγχους 4. Ταυτόχρονα απομακρύνεται το προηγούμενο φορτίο ρύγχους προώθησης, μιας και αυτή η πρώτη περίπτωση φόρτισης ρύγχους της φάσης 139 επιδρά μόνο μέχρι τη CS 135. Δες πίνακα φορτίσεων LC_Tab.
.....	225	C	14	1	Φάση ερπυσμού με χρονική διάρκεια T= 14 ημερών
	301	C	100	1	Φάση ερπυσμού με χρονική διάρκεια T= 100 ημερών μετά την έναρξη της κυκλοφορίας της γέφυρας.
	302	C	30000	5	Φάση ερπυσμού υπολογιζόμενη σε 5 βήματα με χρονική διάρκεια T= 30000 ημερών. Χρόνος άπειρος.

\*\*\* Για τις φάσεις ερπυσμού εφόσον δεν δίνεται η θερμοκρασία περιβάλλοντος και σχετική υγρασία, θα ληφθούν υπόψη οι 'default' τιμές, που είναι: RH=70% και TEMP=20°.

**Πίνακας 7**
**Πίνακας groups στοιχείων**

GRP	NO	ICS1	ATIL	
	0	100	-	Βάθρα
	1	110	-	1η Φάση προώθησης
	2	120	-	2η Φάση προώθησης
	3	130	-	3η Φάση προώθησης
	4	140	-	4η Φάση προώθησης
.....	12	220	-	12η Φάση προώθησης


**Πίνακας 8**

Πίνακας φορτίσεων

LC	NO	ICS1	ATIL		
	100	100	-	Ίδιο βάρος βάθρων	παράγεται αυτόματα
	700	96	105	Ρύγχος προώθησης – Φορτίο 1	δίνεται στο SOFiLOAD
	101	101	-	Περιοχή βάθρων – προένταση	παράγεται αυτόματα
	701	106	115	Ρύγχος προώθησης – Φορτίο 2	δίνεται στο SOFiLOAD
	110	110	-	Ίδιο βάρος φάσης 1	παράγεται αυτόματα
	111	111	-	Φάση 1 – προένταση	παράγεται αυτόματα
	702	116	125	Ρύγχος προώθησης – Φορτίο 3	δίνεται στο SOFiLOAD
	120	120	-	Ίδιο βάρος φάσης 2	παράγεται αυτόματα
	121	121	-	Φάση 2 – προένταση	παράγεται αυτόματα
	703	126	135	Ρύγχος προώθησης – Φορτίο 4	δίνεται στο SOFiLOAD
	.....				
	220	220	-	Ίδιο βάρος φάσης 12	παράγεται αυτόματα
	221	221	-	Φάση 12 – προένταση	παράγεται αυτόματα

Πίνακας 9

## Υπολογισμός παραμορφώσεων – Παράδειγμα cantilever\_simple.dat



Εικόνα 4

Χρησιμοποιώντας ένα πολύ απλοποιημένο παράδειγμα προβολοδόμησης, παρακολουθούμε τον τρόπο με τον οποίο υπολογίζει το πρόγραμμα τις τελικές παραμορφώσεις κόμβων του φορέα που δεν συμμετέχουν από την αρχή στον υπολογισμό του συστήματος. Ανήκουν, δηλαδή, σε κάποιο group στοιχείων που ενεργοποιείται εκ των υστέρων.

Ο υπολογισμός καθορίζεται μέσω της εντολής CTRL, του προγράμματος ASE και την παράμετρο CANT:

**CANT** If new groups and new nodes are activated for instance in cantilevering construction, a primary displacement has to be determined for these new nodes, although they were not still available in the primary load case. This can be controlled with CTRL CANT.

Usage see program CSM Construction Stage Manager.

CANT = 0 no action

CANT = 1 only consideration of displacements

CANT = 2 consideration of displacements and rotations  
= tangential cantilevering construction

CANT = +4 retention of the XY position

Οι τελικές παραμορφώσεις κόμβων του φορέα που δεν συμμετέχουν από την αρχή στον υπολογισμό του συστήματος, μπορούν να υπολογιστούν ως εξής:

- Με αφετηρία την παραμορφωμένη κατάσταση της προϋπάρχουσας εντατικής κατάστασης (PLC), δηλαδή, χωρίς να λαμβάνεται υπόψη μία προϋπάρχουσα παραμόρφωση. Επιλέγουμε CTRL CANT 0.
- Με αφετηρία την απαραμόρφωτη αρχική κατάσταση, λαμβάνοντας υπόψη μία προϋπάρχουσα παραμόρφωση από τις συνολικές μετακινήσεις του φορέα μέχρι τότε. Επιλέγουμε CTRL CANT 1.
- Με αφετηρία την απαραμόρφωτη αρχική κατάσταση, λαμβάνοντας υπόψη μία προϋπάρχουσα παραμόρφωση από τις συνολικές μετακινήσεις και στροφές του φορέα μέχρι τότε. Τότε επιλέγουμε CTRL CANT 2.



Στο συγκεκριμένο παράδειγμα χρησιμοποιούνται δύο φάσεις κατασκευής ιδίου βάρους:

```
#define ASE_CTRL
  ECHO reac,forc,nost,bedd no
  ECHO disp extr
  CTRL DIFF 1000 $ difference loadcases increment
  NSTR S0 KSV SL
  CTRL CANT 2 $ 2 = tangential free cantilever
               $ 1 = with original inclination
#endif
$
#define ASE_SYST=prob line $ nonl nmat ja iter 60
$
$ ----- Construction Stage 10 -----
PROG ASE
HEAD Construction stage CS 10 G1
#include ASE_CTRL
SYST PLC 0 $(ASE_SYST)
GRP 0 CS 10
LC 4010 TYPE G_1 DLZ 1.0 TITL 'G1'
END
$ ----- Construction Stage 20 -----
PROG ASE
HEAD Construction stage CS 20 G1
#include ASE_CTRL
SYST PLC 4010 $(ASE_SYST)
GRP 0 CS 20
GRP 2 CS 20
LC 4020 TYPE G_1 DLZ 1.0 TITL 'G1'
END
```

Αλλάζουμε την τιμή της παραμέτρου, ανάλογα με τον τρόπο που επιθυμούμε να υπολογιστούν οι παραμορφώσεις.

Φάση 10. Ίδιο βάρος στοιχείων group 0. Παράγονται οι φορτίσεις 4010 και 5010.

Φάση 20. Ίδιο βάρος στοιχείων group 0 και 2. Ταυτόχρονα λαμβάνεται υπόψη η φόρτιση 4010 σαν προϋπάρχουσα εντατική κατάσταση. Παράγονται οι φορτίσεις 4020 και 5020.

### CTRL CANT 0

Επιλέγοντας CTRL CANT 0, δεν γίνεται κανένας επιπλέον υπολογισμός, άρα, οι συνολικές παρ/σεις των κόμβων του group 2 είναι αυτές της φόρτισης 5020.

Τα αποτελέσματα του προγράμματος είναι τα ακόλουθα:

Nodal Displacements							
Node.	LC	UX	UY	UZ	PHIX	PHIY	PHIZ
	Nr	Nr	[mm]	[mm]	[mm]	[mrad]	[mrad]
	11	4010	0.085	0.000	0.572	0.000	-0.066
		4020	2.842	0.000	7.278	0.000	-0.878
		5010	0.085	0.000	0.572	0.000	-0.066
		5020	2.757	0.000	6.705	0.000	-0.813
1000	4020	2.830	0.000	10.948	0.000	-0.875	0.000
	5020	2.830	0.000	10.948	0.000	-0.875	0.000

### CTRL CANT 1

Επιλέγοντας CTRL CANT 1, για τον υπολογισμό των παρ/σεων των κόμβων του group 2, εκτιμάται μία προϋπάρχουσα παραμόρφωση, από τις μετακινήσεις της προϋπάρχουσας κατάστασης (Εικόνα 5).

Τα αποτελέσματα του προγράμματος είναι τα ακόλουθα:

Nodal Displacements							
Node.	LC	UX	UY	UZ	PHIX	PHIY	PHIZ
	Nr	Nr	[mm]	[mm]	[mm]	[mrad]	[mrad]
1	11	4010	0.085	0.000	0.572	0.000	-0.066
2		4020	2.842	0.000	7.278	0.000	-0.878
3		5010	0.085	0.000	0.572	0.000	-0.066
4		5020	2.757	0.000	6.705	0.000	-0.813
5	1000	4020	2.915	0.000	11.520	0.000	-0.875
6		5020	2.830	0.000	10.948	0.000	-0.875

Σειρά 1: Αρχικές παρ/σεις κόμβου 11, φάσης 10. Συμπίπτουν με αυτές της φόρτισης 5010 (σειρά 3).

Σειρά 2: Συνολικές παρ/σεις κόμβου 11, φάσης 20. Είναι το άθροισμα των παρ/σεων των φορτίσεων 5010 και 5020 (σειρές 3 και 4 αντίστοιχα).

Σειρά 6: Σχετικές παρ/σεις κόμβου 1000 στην φάση 10. Σε αυτές τις παρ/σεις προστίθενται οι προϋπάρχουσες που προκύπτουν από αυτές του κόμβου 11 (Εικόνα 5), χωρίς να λαμβάνονται υπόψη οι στροφές, ως εξής:



$$U_x = U_x (\text{κόμβου } 1000 - \text{φόρτιση } 5020) + U_x (\text{κόμβου } 11 - \text{φόρτιση } 5010) =$$

$$U_x = 2.830 + 0.085 = 2.915 \text{ mm (σειρά 5)}$$

$$U_z = U_z (\text{κόμβου } 1000 - \text{φόρτιση } 5020) + U_z (\text{κόμβου } 11 - \text{φόρτιση } 5010) =$$

$$= 10.948 + 0.572 = 11.52 \text{ mm (σειρά 5)}$$

Σειρά 5: Συνολικές παρ/σεις κόμβου 1000 στην φάση 20.

### CTRL CANT 2

Επιλέγοντας CTRL CANT 2, για τον υπολογισμό των παρ/σεων των κόμβων του group 2, εκτιμάται μία προϋπάρχουσα παραμόρφωση, από τις μετακινήσεις και τις στροφές της προϋπάρχουσας κατάστασης (Εικόνα 5).

Τα αποτελέσματα του προγράμματος είναι τα ακόλουθα:

#### Nodal Displacements

Node.	LC	UX	UY	UZ	PHIX	PHIY	PHIZ
Nr	Nr	[mm]	[mm]	[mm]	[mrad]	[mrad]	[mrad]
1	11 4010	0.085	0.000	0.572	0.000	-0.066	0.000
2	4020	2.842	0.000	7.278	0.000	-0.878	0.000
3	5010	0.085	0.000	0.572	0.000	-0.066	0.000
4	5020	2.757	0.000	6.705	0.000	-0.813	0.000
5	1000 4020	2.915	0.000	11.848	0.000	-0.940	0.000
6	5020	2.830	0.000	10.948	0.000	-0.875	0.000

Σειρά 1: Αρχικές παρ/σεις κόμβου 11, φάσης 10. Συμπίπτουν με αυτές της φόρτισης 5010 (σειρά 3).

Σειρά 2: Συνολικές παρ/σεις κόμβου 11, φάσης 20. Είναι το άθροισμα των παρ/σεων των φορτίσεων 5010 και 5020 (σειρές 3 και 4 αντίστοιχα).

Σειρά 6: Σχετικές παρ/σεις κόμβου 1000 στην φάση 10. Σε αυτές τις παρ/σεις προστίθενται οι προϋπάρχουσες που προκύπτουν από αυτές του κόμβου 11 (Εικόνα 5), λαμβάνοντας υπόψη και τις στροφές, ως εξής:


$$U_x = U_x (\text{κόμβου } 1000 - \text{φόρτιση } 5020) + U_x (\text{κόμβου } 11 - \text{φόρτιση } 5010) =$$

$$U_x = 2.830 + 0.085 = 2.915 \text{ mm (σειρά 5)}$$

$$U_z = U_z (\text{κόμβου } 1000 - \text{φόρτιση } 5020) + U_z (\text{κόμβου } 11 - \text{φόρτιση } 5010) - 5\text{m} * \tan \phi_{11} (\text{κόμβου } 11) =$$


$$= 10.948 + 0.572 - 5 * \tan (-0.066) = 11.850 \text{ mm}$$

Σειρά 5: Συνολικές παρ/σεις κόμβου 1000 στην φάση 20.



Εικόνα 5

Προκ/μένη γέφυρα με σκυροδέτηση επιτόπου (in-situ) – Παράδειγμα  
csm3\_stab\_fertigteil\_ob.dat



Εικόνα 6

Η διατομή αποτελείται από δύο τμήματα. Το πρώτο είναι το προκατασκευασμένο τμήμα και ενεργοποιείται στην φάση κατασκευής 1 και το δεύτερο είναι το τμήμα που σκυροδετείται επιτόπου στην φάση 40. Αυτές οι φάσεις ορίζονται από την αρχή στο πρόγραμμα AQUA, μέσω της εντολής CS:

SECT 1 MNO 1 MRF 2

CS 1

POLY OPZ MNO 1

VERT 1 1.125 0.000

2 1.125 0.110

3 0.300 0.150

4 0.275 1.050

Κατά την φάση CS 1 (αρχή των χρόνων), ενεργοποιείται το προκατασκευασμένο μέρος της διατομής.

CS 40

POLY OPZ MNO 4

VERT 11 1.125 -0.20


12 1.125 0.00

Κατά την φάση CS 40, ενεργοποιείται το επιπλέον μέρος της διατομής. Από την CS 40 και μετά η διατομή είναι πλήρης.

CROSS-SECTIONS STATIC PROPERTIES

No	MNo	A[m2]	Ay/Az/Ayz	Iy/Iz/Iyz	ys/zs	y/z-sc	modules	gam
	MNs	It [m4]	[m2]	[m4]	[m]	[m]	[MPa]	[kN/m]
1	1	1.2281E+00		1.573E-01	0.000	0.000	31387	31.80
	2	8.378E-02		3.028E-01	0.234	0.075	13078	
1.1	= CS 1							
	1	8.2200E-01		8.827E-02	0.000	0.000	31387	20.55
	2	4.652E-02		1.316E-01	0.399	0.210	13078	
1.2	= CS 40							
	1	1.2281E+00		1.573E-01	0.000	0.000	31387	31.80
	2	8.378E-02		3.028E-01	0.234	0.075	13078	

Cross section No. 1



Εικόνα 7

Τα δεδομένα του προγράμματος GEOS είναι τα ακόλουθα:

PROG GEOS

HEAD Παραβολική προένταση

§ ΓΕΩΜΕΤΡΙΑ

CBEA NOH 1 FROM 10002 10007 TYPE BEAM

CBEA NOH 2 FROM 10010 10015 TYPE BEAM

```

T GEO NOG 1 NOH 1      'Τένοντας 1'
PTSC NO  X      U      V      DVS
10002 0.00      0.00  0.40  -
10005 0.00      =      0.95  0
10007 -.00001      =      0.40  -
T GEO NOG 2 NOH 2      'Τένοντας 2'
PTSC NO  X      U      V      DVS
10010 0.00      0.00  0.40  -
10013 0.00      =      0.95  0
10015 -.00001      =      0.40  -
$ ΜΕΘΟΔΟΣ ΠΡΟΕΝΤΑΣΗΣ
PRES NOT 1 NOG 1 ICS1 11 12 0 TITL 'Τένοντας 1'
PSYS MNO 1011 ZV 1635 AZ 1680 195000 4.8 0.3 0.19 0.19 SS 3 DH 75 TITL B+B-L12
PFAC RI 9
TEND NTEN 2
$
PRES NOT 2 NOG 2 ICS1 11 12 0 TITL 'Τένοντας 2'
TEND NTEN 2
$ ΑΠΟΘΗΚΕΥΣΗ ΦΟΡΤΙΣΗΣ ΠΡΟΕΝΤΑΣΗΣ ΣΤΗΝ ΒΑΣΗ ΔΕΔΟΜΕΝΩΝ
LOAD NOT 1,2 LC 11
$ -- BLOCK: GRAF Γραφική παράσταση γεωμετρίας καλωδίων
ECHO PLOT FULL
SIZE TYPE 4 SC 0
SHOW NO 1 TYPE GEOE FACH 4 TYPG DUCT DIA 65 COLO -1 PCS 1 ; AND
SHOW NO 1 TYPE GEOE FACH 4 TYPG TEND
SHOW 1 TYPE FACT FACH 50 PCS 1
END

```

Περιγράφονται οι τένοντες προέντασης, που ενεργοποιούνται στην φάση 11 (ICS1=11) και αποθηκεύονται στην φόρτιση LC 11.

Πίνακας φάσεων κατασκευής

CS	NO	TYPE	T(d)	NCRE	
	10	G_1	-		Σκυροδέτηση των προκατ/ων τμημάτων (συνθήκες στήριξης τμήματος εργοστασίου).
	11	P	-	1	Προένταση προκατ/ων τμημάτων.
	15	C	40		Φάση ερπυσμού με χρονική διάρκεια T= 40 ημερών.
	18	B	-		Βοηθητικές στηρίξεις κατά τη κατασκευή.
	20	G_1	-		Επιτόπου Σκυροδέτηση, ενεργοποίηση ι.β. νωπού σκυροδέματος, χωρίς τη πλήρη ακαμψία του – βλ. GRP SITU.
	40	B	-		Σκλήρυνση 'επί τόπου' τμήματος διατομής. Απομάκρυνση βοηθητικών στηρίξεων. Ενεργοποίηση τελικής θεμελίωσης.
	50	G_2	-		Από δω και κάτω το 'επιτόπου' σκυρόδεμα έχει πλήρη ακαμψία.
	55	C	50		Ασφαλτικά, επικαλύψεις.
	65	C	100		Φάση ερπυσμού μέχρι τη παράδοση της γέφυρας στη κυκλοφορία.
	75	C	30000	5	Φάση ερπυσμού μετά την επιβολή των φορτίων κυκλοφορίας.
					Φάση ερπυσμού μέχρι t-άπειρο (φάσεις ερπυσμού με delta-phi > 0.4 πρέπει να μοιραστούν σε περισσότερα τμήματα NCRE).

\*\*\* Για τις φάσεις ερπυσμού εφόσον δεν δίνεται η θερμοκρασία περιβάλλοντος και σχετική υγρασία, θα ληφθούν υπόψη οι 'default' τιμές, που είναι: RH=70% και TEMP=20°.

Πίνακας 10

Πίνακας Groups στοιχείων

GRP	NO	ICS1	ATIL	SITU*	T0	
	(0 39 1)	10	-	20	7	Προκ/νες δοκοί
	40	40	-	-	7	Ενδιάμεσα ραβδωτά στοιχεία που εξασφαλίζουν την συνέχεια στις προκ/νες δοκούς.
	91	10	17	-	-	Προσωρινή έδραση προκ/μένου τμήματος
	92	18	39	-	-	Προσωρινή έδραση κατά την σκυροδέτηση
	93	40	-	-	-	Τελική έδραση.

(\*) SITU 20 σημαίνει εδώ, ότι για τα στοιχεία των groups 0 έως 39, στην CS 20, πρέπει να ενεργοποιηθεί το ίδιο


βάρος του νωπού σκυροδέματος για το νέο τμήμα των διατομών αυτών. Η πλήρης ακαμψία της διατομής ενεργοποιείται στην CS 40 (AQUA...SECT... με CS 40).

Πίνακας 11

Η εντολή που παράγεται στα δεδομένα του προγράμματος ASE, για τον υπολογισμό του ίδιου βάρους του νωπού σκυροδέματος, στην φάση 20 είναι:

GRP 10 CS 20 CSDL 999

Η παράμετρος CSDL ενεργοποιεί το ίδιο βάρος τμημάτων της διατομής επόμενων φάσεων (εδώ 999, δηλ. σε φάση μεγαλύτερη ή ίση με την 40). Θα πρέπει πάντα να γίνεται ένας έλεγχος των φορτίων που δρουν στην φόρτιση 4020, μέσω του προγράμματος WinGRAF, για να διαπιστώνεται ότι το ίδιο βάρος που ενεργοποιείται είναι αυτό της πλήρους διατομής.




Εικόνα 8

Πίνακας φορτίσεων

LC	NO	TYPE	ICS1	ATIL	
LC	2	G 2	50	-	Πρόσθετα μόνιμα φορτία λειτουργίας G2.

Πίνακας 12

Προεντεταμένη κλίνη – Παράδειγμα csm3\_spannbett-eng.dat



Εικόνα 9


Τροποποιώντας τα δεδομένα του παραδείγματος csm3\_stab\_fertigteil\_ob.dat, παίρνουμε τη περίπτωση της προεντεταμένης κλίνης. Η διατομή συμπληρώνεται σε δύο φάσεις, όπως και προηγουμένως.

```

PROG AQUA
HEAD CSM - Δοκός δύο ανοιγμάτων με προκ/νες δοκούς και επί τόπου σκυροδέτηση
ECHO SECT FULL
NORM DIN FB-102 CAT C      $ DIN FACHBERICHT 102 βλέπε παράδειγμα DIN_FB_SOFiSTiK
CONC 1 C 40                $ = C40/50
STEE 2 BST 500SA TITL 'Χαλαρός οπλισμός'
CONC 4 C 30                $ = C30/37 In situ beton
STEE 11 PST 1570S

SECT 1 MNO 1 MRF 2
let#zmax 1.05
CS 1
POLY OPZ MNO 1
VERT FOMI 0.000 0.000 AIR 0.10 $ κυκλοφορία αέρα μόνο πάνω από την ασφαλτο
1 1.125 0.000 AIR 1.00 $ για όλα τα υπόλοιπα σημεία VERT.
2 1.125 0.110
3 0.300 0.150
4 0.275 #zmax
CS 40
POLY OPZ MNO 4
VERT OOMI 0.000 -0.200 AIR 0.40 $ κυκλοφορία αέρα μόνο πάνω από την ασφαλτο
VERT 11 1.125 -0.200 AIR 1.00
VERT 12 1.125 0.000 AIR 1.00
VERT OUMI 0.000 0.000 AIR 0.20 $ κυκλοφορία αέρα μόνο ως πιο πάνω

```



Εικόνα 10

Μέσω της παραμέτρου AIR της εντολής VERT, μπορεί να καθοριστεί ο βαθμός έκθεσης της περιμέτρου μίας διατομής στη ξήρανση επιφάνειας σκυροδέματος. Δίνοντας στη παράμετρο AIR τιμές στις κορυφές της περιμέτρου κάποιας διατομής, τότε ο βαθμός έκθεσης της κάθε πλευράς της περιμέτρου είναι ο γεωμετρικός μέσος όρος των τιμών των κορυφών. Π.χ. εάν σε δύο κορυφές έχουν δοθεί οι τιμές 0.1 και 1.00, τότε ο βαθμός έκθεσης της πλευράς που καθορίζεται από τις κορυφές αυτές είναι ίσος με  $\sqrt{0.1 \cdot 1.00} = 0.316$ . Η τιμή αυτή λαμβάνεται υπόψη στον υπολογισμό του ενεργού πάχους του στοιχείου  $deff = 2 \cdot A/u$ , από το οποίο εξαρτάται ο συντελεστής ερπυσμού του συγκεκριμένου στοιχείου. Α είναι το συνολικό εμβαδόν της διατομής, u είναι η περίμετρος της διατομής που εκτίθεται στη ξήρανση επιφάνειας σκυροδέματος.

Στην διατομή στην Εικόνα 10 ισχύουν:

$A(\text{υλικού } 1) = 0.822\text{m}^2$ ,  $u(\text{υλικού } 1) = 6.473\text{m}$

Βαθμός έκθεσης περιμέτρου τμήματος με υλικό 1:  $\Sigma\{(\text{βαθμός έκθεσης πλευράς}) \cdot (\text{μήκος πλευράς})\} / (\text{περίμετρο}) = \sqrt{(0.1 \cdot 1.00) \cdot 2.25 + 1.00 \cdot 4.223} / 6.473 = 0.7623$

$$\text{Αρα, } deff = \frac{2 \cdot 0.822E04}{0.7623 \cdot 6.473E02} = 33.32\text{cm}$$

#### Cross-sections static properties

No	MNo	A[m2]	Ay/Az/Ayz	Iy/Iz/Iyz	ys/zs	y/z-sc	modules	gam
	MNs	It[m4]	[m2]	[m4]	[m]	[m]	[MPa]	[kN/m]
1.1	=	CS 1						
	1	8.2200E-01		8.827E-02	0.000	0.000	31387	20.55

2 4.652E-02 1.316E-01 0.399 0.210 13078

## Additional Design Data

M	periphery-O/-I	deff	t-min	t-max	SMP	thet-p	thet-y	thet-z	thet-yz
	[m2/m]	[cm]	[cm]	[cm]	[o/o]	[tm2/m]	[tm2/m]	[tm2/m]	[tm2/m]
1	6.473	33.32			0.0	5.496	2.207	3.289	0.000


Τα δεδομένα του προγράμματος GEOS είναι τα ακόλουθα:

```

PROG GEOS
HEAD Άμεση συνάφεια με μήκος αδράνειας - δεξ PSYS FTL
$ ΓΕΩΜΕΤΡΙΑ
  CBEA NOH 1 FROM 2 8 TYPE AUTO ; TOPP 1 1 KIND NODE 2 ; TOPP 1 2 KIND NODE 8
  CBEA NOH 2 FROM 10 16 TYPE AUTO ; TOPP 2 1 KIND NODE 10 ; TOPP 2 2 KIND NODE 16
  TGEO NOG 1 NOH 1 'Τένοντας 1'
  PTSC NO X U V
        0 0.0,1.5,2.0 0.00 0.82 $ NO=0 = xi-τιμές με xi = 0, 1.5 και 2.0
  TGEO NOG 2 NOH 2 'Τένοντας 2'
  PTSC NO X U V DVS
        0 0.0,1.5,2.0 0.00 0.82
$ ΜΕΘΟΔΟΣ ΠΡΟΕΝΤΑΣΗΣ ΑΜΕΣΗΣ ΣΥΝΑΦΕΙΑΣ
  PRES NOT 1 NOG 1 ICS1 0 0 0 TITL 'Τένοντας 1'
  PSYS MNO 1011 ZV 3*1635 AZ 3*1680 MUE 0 0 SS 0 DH 0.0 FTL 3.0 FTYP LINE
  PFAC RI 1 $ Το αποτέλεσμα από FTL 3.0 φαίνεται από το
  TEND NTEN 1 $ διάγραμμα αξονικών δυνάμεων στο csm3_spannbett_csm.plb
  PRES NOT 2 NOG 2 ICS1 0 0 0 TITL 'Τένοντας 2'
  TEND NTEN 1
$ ΑΠΟΘΗΚΕΥΣΗ ΦΟΡΤΙΣΗΣ ΠΡΟΕΝΤΑΣΗΣ ΣΤΗΝ ΒΑΣΗ ΔΕΔΟΜΕΝΩΝ
  LOAD NOT 1,2 LC 11
$ -- BLOCK: GRAF Γραφική παράσταση γεωμετρίας καλωδίων
  ECHO PLOT FULL
  SIZE TYPE 4 SC 0
  SHOW NO 1 TYPE GEOE FACH 4 TYPG DUCT DIA 65 COLO -1 PCS 1 ; AND
  SHOW NO 1 TYPE GEOE FACH 4 TYPG TEND
  SHOW 1 TYPE FACT FACH 50 PCS 1
END

```

Δίνοντας στην εντολή PRES, στις παραμέτρους ICS1 και ICS2 τη τιμή 0, οι δύο τένοντες αποκτούν άμεση συνάφεια. Την γεωμετρία των τενόντων (ευθύγραμμοι) τη περιγράφουμε με τουλάχιστον τρία υποχρεωτικά σημεία, μέσω της εντολής PTSC. Επιπλέον χαρακτηριστικό της προεντεταμένης κλίνης είναι ότι ο συντελεστής τριβής και η ολίσθηση είναι 0 (MUE=0 SS=0). Το μήκος αδράνειας περιγράφεται μέσω της εντολής PSYS, παραμέτρου FTL (εδώ ίσο με 3m).



Εικόνα 11

Ανεξάρτητα από το γεγονός ότι οι τιμές των ICS1 και ICS2 είναι ίσες με μηδέν, φροντίζουμε ο αριθμός της φόρτισης στην οποία θα αποθηκευθεί η προένταση να ταυτίζεται με τον αριθμό της φάσης στην οποία θα ενεργοποιηθεί (LC 11).

Μέσω του προγράμματος SOFiLOAD, καθορίζονται τα πρόσθετα φορτία, καθώς και οι δράσεις που θα χρησιμοποιηθούν κατά την δημιουργία των συνδυασμών για εύρεση περιβαλλουσών εντατικών μεγεθών.

```

PROG SOFiLOAD
HEAD Definition of loadcase G2 in final system
ACT G
ACT G_1
ACT G_2
ACT P GAMU 1.00 GAMF 1.00 SUP PERM
ACT K GAMU 1.00 GAMF 1.00 SUP PERM
ACT B GAMU 1.35 GAMF 1.00 SUP PERM PSIO 1.00 PSI1 1.00 PSI2 1.00
ACT X GAMU 1.35 GAMF 1.00 $ Βοηθητικές στηρίξεις
$

```

LC 2 TYPE G\_2 TITL G\_2 ; BEAM GRP 10,40 TYPE PZP PA 20  
END

Πίνακας φάσεων κατασκευής

CS	NO	TYPE	T(d)	NCRE	
	10	G_1	-		Σκυροδέτηση των προκατ/ων τμημάτων
	11	P	-		Προένταση προκατ/ων τμημάτων. Υποχρεωτική η LC 11.
	15	C	40	1	Φάση ερπυσμού με χρονική διάρκεια T= 40 ημερών.
	18	B	-		Βοηθητικές στηρίξεις κατά τη κατασκευή.
	20	G_1	-		Επιτόπου Σκυροδέτηση, ενεργοποίηση ι.β. νωπού σκυροδέματος, χωρίς τη πλήρη ακαμψία του – βλ. GRP SITU.
	40	B	-		Σκλήρυνση 'επί τόπου' τμήματος διατομής.
	50	G_2	-		Από δω και κάτω το 'επιτόπου' σκυρόδεμα έχει πλήρη ακαμψία.
	55	C	50		Ασφαλτικά, επικαλύψεις.
	65	C	100		Φάση ερπυσμού μέχρι τη παράδοση της γέφυρας στη κυκλοφορία.
	75	C	30000	5	Φάση ερπυσμού μετά την επιβολή των φορτίων κυκλοφορίας.
					Φάση ερπυσμού μέχρι t-άπειρο (φάσεις ερπυσμού με delta-phi > 0.4 πρέπει να μοιραστούν σε περισσότερα τμήματα NCRE).

\*\*\* Για τις φάσεις ερπυσμού εφόσον δεν δίνεται η θερμοκρασία περιβάλλοντος και σχετική υγρασία, θα ληφθούν υπόψη οι 'default' τιμές, που είναι: RH=70% και TEMP=20°.

Πίνακας 13

Πίνακας Groups στοιχείων

GRP	NO	ICS1	ATIL	SITU*	T0	
	10	10	-	20	7	Προκ/νες δοκοί
	40	40	-	-	7	Ενδιάμεσα ραβδωτά στοιχεία που εξασφαλίζουν την συνέχεια στις προκ/νες δοκούς.
	91	10	17	-	-	Προσωρινή έδραση προκ/μένου τμήματος
	92	18	47	-	-	Προσωρινή έδραση κατά την σκυροδέτηση
	93	48	-	-	-	Τελική έδραση.

(\*) SITU 20 σημαίνει εδώ, ότι για τα στοιχεία του group 10, στην CS 20, πρέπει να ενεργοποιηθεί το ίδιο βάρος του νωπού σκυροδέματος για το νέο τμήμα των διατομών αυτών. Η πλήρης ακαμψία της διατομής ενεργοποιείται στην CS 40 (AQUA...SECT... με CS 40).

Πίνακας 14

Πίνακας φορτίσεων

LC	NO	TYPE	ICS1	ATIL	
	2	G_2	50	-	Πρόσθετα μόνιμα φορτία λειτουργίας G2.

Πίνακας 15

Επειδή το υλικό 4 ενεργοποιείται αργότερα (φάση 40), πρέπει μέσω της εντολής CREP να διαφοροποιηθούν οι συνολικοί συντελεστές ερπυσμού και συστολής ξήρανσης:

CREP MNO 4 GRP 40 PHI 1.88 EPS -30E-5